const TITLE = "Самый тестовый макрос в мире....";

macro ovb_s2date( val, dibug ) : date	//di через i написано умышленно 

	var	err = "";

//mod (val1:variant, val2:variant) :variant 
//Процедура возвращает остаток от деления величины val1 на величину val2. 
//
//Параметры: 
//В качестве параметров процедуры используются значения типаDouble иInteger. 

	macro isVisokosniy( y ) : bool
		var	ret;
		if( (mod(y,4)==0) and ((mod(y,100)!=0)) )		return true;
		end;
		if( (mod(y,400)==0) )				return true;
		end;
		return false;
	end;
	macro isLegalPars( str ) : bool
		var	n, m, y;
		var	TNM = ",2,";
		var	SHM = ",4,6,9,11,";
		var	LGM = ",1,3,5,7,8,10,12,";
		var	sm, nFEB = 28;

		if( substr(str,3,1) != "." )	err="lp1";		return false;		end;
		if( substr(str,6,1) != "." )	err="lp2";		return false;		end;
		if( strlen(str)     !=  10 )	err="lp3";		return false;		end;
		m = int(substr(str,4,2)); // месяц
		if( m <=  0 )				err="lp6";		return false;		end;
		if( m  > 12 )				err="lp7";		return false;		end;
		sm = ","+string(m)+",";

		y = int(substr(str,7,4));
		if( y <= 1920 )			err="lp8";		return false;		end;
		if( y  > 2100 ) 			err="lp9";		return false;		end;

		if( isVisokosniy(y) )	nFEB = 29;
		end;

		n = int(substr(str,1,2));
		if( n <=  0 )				err="lp4";		return false;		end;
		if( (index(TNM, sm) > 0)
			and (n > nFEB) )		err="lp5.T";		return false;		end;
		if( (index(SHM, sm) > 0)
			and (n > 30) )		err="lp5.S";		return false;		end;
		if( (index(LGM, sm) > 0)
			and (n > 31) )		err="lp5.L";		return false;		end;

		return true;
	end;

	macro isDigit( s )
		if( codefor(string(s)) < 48 )	return false;				end;
		if( codeFor(string(s)) > 58 )	return false;				end;
		return true;
	end;
	macro 	_ssm( str, mon, num ) : string
		macro firstNoneLetter( str ):integer
			var i=0;
			while( not isDigit( substr(str,i+1,1)) )
			   i=i+1;
			end;
			return i;
		end;

		var n,t;
		if( index( str, mon ) == 0 )	return str;
		end;
		n = index(str, mon);
		//t = index( substr(str,n+1),"#" );
		t = firstNoneLetter( substr(str,n+1) );
		str=substr(str,1,n-1) + "#" + num + "#" + substr(str,t+n+1);
		return str;
	end;
	macro letcut( str, n ):string
		while( isDigit( substr(str,n,1)))
			n=n+1;
		end;
		return substr(str,1, n-1);
	end;

	var ret = date(0,0,0);
	var str = val;
	if( valtype(dibug) == V_UNDEF )	dibug = false;
	end;
	if ( valtype(val) == V_DATE )
	   [ret ##########   str:############################### val ############################]
	   (string(val), str, val);
          return val;
	end;
	
	if(dibug)[str:###############################](str);end;
	str = trim( str );
	str = StrLwr( str );
	str = strsubst( str, "-", "#");
	str = strsubst( str, "/", "#");
	str = strsubst( str, ".", "#");
	str = strsubst( str, "_", "#");
	str = strsubst( str, " ", "#");
	if(dibug)[str:###############################](str);end;
	str = strsubst( str, "#####", "#");
	str = strsubst( str, "####" , "#");
	str = strsubst( str, "###"  , "#");
	str = strsubst( str, "##"   , "#");
	if(dibug)[str:###############################](str);end;

	str = strsubst( str, "года#", "");
	str = strsubst( str, "#г."  , "");
	str = strsubst( str, "#года", "");
	str = strsubst( str, "#год" , "");
	if(dibug)[str:###############################](str);end;

	str = strsubst( str, "#г#"  , "");
	str = strsubst( str, "г#"   , "");
	str = strsubst( str, "года" , "");
	str = strsubst( str, "г"    , "");
	str = strsubst( str, "года.", "");
	if(dibug)[str:###############################](str);end;

	str = _ssm( str, "ян",  "01" );
	str = _ssm( str, "фе",  "02" );
	str = _ssm( str, "мар", "03" );
	str = _ssm( str, "ап",  "04" );
	str = _ssm( str, "ма",  "05" );
	str = _ssm( str, "июн", "06" );
	str = _ssm( str, "июл", "07" );
	str = _ssm( str, "ав",  "08" );
	str = _ssm( str, "се",  "09" );
	str = _ssm( str, "ок",  "10" );
	str = _ssm( str, "но",  "11" );
	str = _ssm( str, "д",   "12" );

	str = strsubst( str, "#####", "#");
	str = strsubst( str, "####" , "#");
	str = strsubst( str, "###"  , "#");
	str = strsubst( str, "##"   , "#");
	if(dibug)[str:###############################](str);end;

	if( index(str,"#" ) == 2 ) 		str = "0" + str;
	else					err = "i2";
	end;
	if( substr(str,3,1) == "#" ) 	strset( str, 3, ".");
	else					err = "s2";
	end;
//03.4#16
	if( index(str,"#" ) == 5 ) 		str = substr(str,1,3) + "0" + substr(str,4);
	else					err = "s5";
	end;
//03.04#16
	if( index(str,"#" ) == 6 ) 		strset( str, 6, ".");
	else					err = "i6";
	end;

	str = letcut( str, 7 );
	if(dibug)[letcut> str:###############################](str);end;


//03.04.16
	if( strlen(str) == 8 ) 		str = substr(str,1,6) + "20" + substr(str,7);
	else					err = "s8";
	end;
	if(dibug)[str:###############################](str);end;

//31.12.2016
	if( not isLegalPars( str ) )	//msgbox("err5901: ошибка при разборе даты: ", val );
		[err5901: ошибка при разборе даты:######### str:############################### val ############################################ ret ##########]
		(err, str, val, string(ret));
		return ret;
	end;
//30.02.2016
//	if( 

	ret = date( str );
//	[ret ###############################################        ##########    ###########        ]
//	(val, string(ret), str);
	[ret ###############################################        ################]
	(val, string(ret));
	return ret;
end;





macro main( ) : bool
//msgbox("main");
	var	d = "3апреля2016г.";
	var	n = "";

//while( getstring(d, "было: "+d+"\nстало: "+n+"\n\nВведите новую кривую дату", 40) )
//		n = ovb_s2date(d);
//		n = string(n);
//end;
//return;


	d = ovb_s2date("3апреля16годааааааа");
	d = ovb_s2date("3 апрЮляДрюля дляхамуха 2016вот такая шняга");
	d = ovb_s2date(" 28 02 2017 ");
	d = ovb_s2date(" 3 май   2016 ");
	d = ovb_s2date(" 03.04.2016 ");
	d = ovb_s2date(" 03-04-2016 ");
	d = ovb_s2date("    			  03.04.16     ");
	d = ovb_s2date("3-04.2016			");
	d = ovb_s2date("3.4.2016");
	d = ovb_s2date("29 / 4   		/ 2016");
	d = ovb_s2date("3.4.16");
	d = ovb_s2date(" 3   4   16   ");
	d = ovb_s2date("3 апреля 2016");
	d = ovb_s2date("5мая16");
	d = ovb_s2date("5июня2016");
	d = ovb_s2date("19 апреля1963");
	d = ovb_s2date("9 май1963");
	d = ovb_s2date("19 июня 63");
	d = ovb_s2date("23 июл16года");
	d = ovb_s2date("23авг16года.");
	d = ovb_s2date("23сент16г.");
	d = ovb_s2date("23/окт/16г");
	d = ovb_s2date("23/ноярбя/2016 года");
	d = ovb_s2date("31/декарб2016года");
	d = ovb_s2date("3 апр. 16 года");
	d = ovb_s2date("3 апреля 2016 года");
	d = ovb_s2date("3 апреля 2016 г.");
	d = ovb_s2date("3 апреля 2016года");
	d = ovb_s2date("3 янв 2016года");
	d = ovb_s2date("31 дек 2016года.");
	d = ovb_s2date("31 дикарбя 2016года.");
	d = ovb_s2date("14.06.1986г");

[ Ошибочные примеры, тут прога оказалась бессильной];
	d = ovb_s2date(" 03.24.3016 ");
	d = ovb_s2date(" 53 14   2016 ");
	d = ovb_s2date(" 31 фев 2016 ");
	d = ovb_s2date(" 33 мар 2016 ");
	d = ovb_s2date(" 31 апр 2016 ");

end;


main();
